UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

Coal-seismic, desktop computer programs in BASIC; part 8: pick first arrivals and align events on screen

by

Wilfred P. Hasbrouck[1]

Open-File Report 84-175

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS. Although the programs in this report have been extensively tested, the USGS makes no guarantee whatever of correct results.

[1] Golden, Colorado

1984

ABSTRACT

Processing of geophysical data taken with the U.S. Geological Survey's coal-seismic system is done with a desktop, stand-alone computer. Programs for this computer are written in the extended BASIC language used by the Tektronix 4051 Graphic System. This report presents computer programs to automatically pick first arrivals and to align events on the screen to increase the precision of their timing.

INTRODUCTION

The first program of this report presents a computer procedure for automatic picking of first arrivals (first breaks) on seismic records. If speed is the principal factor, then the computer excells; however, if the first arrivals are critical to the problem and if the arrivals are low in amplitude and buried in noise, then judgment of an experienced person must be called upon.

One interactive first-break-picking procedure that can be used is to make initial determinations of first break times automatically, display trace segments on the screen so that they are aligned at their first break times, and then shift arrivals forward or backward in time by entering shift values from the keyboard until a visually satisfactory alignment is obtained. Another procedure, but one that is less precise, is to let the computer make the first break picks and then either accept them without question or examine--and perhaps alter--the tabulated results before acceptance.

It is well known that the onset time of the first pulse arriving at a detector is not defined (Anstey, 1977, p. 2-87)--it can be moved forward or backward in time by changing the amplitude of display (Ricker, 1953, p. 20). Thus, no computer program can be expected to produce highly accurate first arrival times. At best, one can strive through computer assistance to better the accuracy of determination. Because of the high resolution required in shallow-coal seismics, it is essential that static variations (constant time shifts between traces on a seismic record) be removed before stacking proceeds, otherwise the higher frequency arrivals will be smeared and thus lost. If corrections for these static shifts are to be obtained by refraction methods, it follows that first arrival times used in the refraction procedures should be determined precisely.

Two methods commonly are employed to produce the required precision: timing the first trough or peak of the refraction arrival (Anstey, 1977, p. 2-87) or applying the "intercept first kick" procedure of Ricker (1953, p. 21). The automatic first break picking program of this report operates on a first trough/peak determination. Experience with this program forces the warning that blind faith in its results (or those of any automatic procedure) is misplaced.

Although the principal purpose of the event alignment programs--the second and third programs of this report--is to offer a method for both checking and increasing the precision of first break picks, they also can be used to increase the precision of timing reflection arrivals (for example, to determine times for X-square/T-square analysis), to time arrivals obtained in downhole velocity surveys, to identify reflections in the neighborhood of the critical distances, and to pick crossing-dip reflections.

The data processing procedures of this report are part of the U.S. Geological Survey's coal-seismic system. Computer programs are written in an extended BASIC language developed by Tektronix, Inc. for use with their 4051 Graphic System. The program requires four pieces of Tektronix equipment: a 4051 Graphic System with a 32K-byte memory, a 4924 digital cartridge tape drive, a 4631 Hard Copy unit, and a Data Processing ROM. In addition, a special ROM (discussed in the last section) is used.

All programs are self-prompting. In tracing through a sample problem, you will notice that the programs print questions and requests followed by a flashing question mark. The computer then waits for you to enter a response from the keyboard. Replies entered in order to run the sample problems are enclosed in boxes on the figures of this report.

Figure 1a shows the full record used in illustrating the programs of this report. The segment of this record between times of 15 and 65 msec is shown on figure 1b. This record was obtained with the use of a hammer-impact shear source and horizontal seismometers oriented transverse to the spread. On these displays, for each trace the offset (O/S) distance in meters and the position number (PN) of each detector are listed on the right. The position number of the subsurface point (SS), taken as the midpoint between the source and detector, for each trace is tabulated on the left. Printed across the top of the display are the header of the record (13540024), the traces plotted (1 thru 12), the record number (#4), the code name of the master data tape (MDT) on which the record is stored (MDT AR2UT), the sample interval used in the display (1 msec on the upper display and 0.5 msec on the lower display), and the amplitude multiplier (x2 on the upper display and x5 on the lower display) applied to the plotted data.

Inspection of the displays of figures 1a and 1b shows that no first arrivals exist earlier than the time to the left of a sloping straight line beginning at a time of 26 msec on trace 1 and extending to a time of 41 msec on trace 12. Further, on all traces a full wavelength of first arrival events appears to be contained within a 30-msec interval extending beyond the estimated minimum first break times.

AUTOMATIC FIRST BREAK PICKING PROGRAM

For each of the 12 traces of a record stored on a master data tape, the automatic first break picking program produces the arrival time of the first peak or trough beyond the time when the absolute amplitude of the trace exceeds an established set of one to three threshold amplitudes (the first extrema time) and the time following the first peak or trough at which the trace crosses the mean value of the pre-first break noise (the cross-over time). The program then tabulates for each trace the two first break values together with the sample standard deviation of the pre-first break noise. The user is then offered the option to choose and store one set of first extrema and cross-over times on a tape. Stored values can be used as data input to the second of the event alignment programs--third program of this report.



Figure 1a. Test-case seismic record used to illustrate programs.



Figure 1b. Segment of test-case seismic record from 15 to 65 msec.

The signal-enhancement seismograph used in our work has a tendency to introduce minor dc shifts to the trace amplitudes; therefore, the mean value of a trace may not be zero. Also the instrument may produce an unwanted set of crossfed excursions in the early (0 to 20 msec) part of the seismic record. A close look at the record displayed on figure 1a would reveal that we have nulled this early instrumentation noise from 0 to 15 msec. In evaluating the noise level on the no-signal segment of the trace (those times before the first arrival), we do not want to include the nulled interval. We call the time at the beginning of the pre-first break noise window the "pre-fb start time"--in the example to follow (fig. 3) this time is 15 msec.

Upon receiving input information as entered upon responses to prompts (see fig. 3), the program follows the computational path outlined below, (reference to fig. 2 is made in this discussion):

1. Finds, retrieves, converts (from hexadecimal to decimal), and scales data stored on a master data tape (MDT AR2UT, fig. 1).

2. Establishes minimum first break times for traces 2 through 11 using linear interpolation between the entered end-trace values--26 msec on trace 1; 41 msec on trace 12.

3. Determines the sample standard deviation and the mean value for data contained within the pre-first break noise window for each trace--this window begins at the pre-fb start time (15 msec) and ends at the minimum first break time (26 msec) in the example on figure 2.

4. Finds the time (T3 on fig. 2) on each trace at which the absolute amplitude exceeds an amplitude equal to a multiple of the sample standard deviation of values within the pre-first break noise window--we call these "thresholds". In the example shown on figure 3, the first threshold multiplier is 2 and the number of threshold multipliers is 3. Since for the example shown in figure 2 (early segment of trace 1 of the record shown on fig. 1) the sample standard deviation for its pre-fb noise window is 6.5, the first threshold value is 2 x 6.5 = 13, the second is 19.5, and the third is 26. The corresponding T3 values (to the nearest sample interval time, 0.5 msec) are 29.5, 30, and 30.5 msec respectively for these three thresholds. This use of incremented threshold multipliers was suggested by James H. Scott (U.S. Geological Survey, oral commun., 1983).



Figure 2. Sketch to illustrate terms used in the automatic first break picking program.

5. Locates the time beyond T3 at which the trace crosses the mean value of pre-first break noise for the second time; for example, if the recordings were made such that first break motion was displayed as down-going, then this second cross-over time (T7, fig. 2) would be that time at which those trace values after the first peak following the first trough cross over the mean value of the pre-first break noise level.

6. Smooths data contained within the T3/T7 interval using a seven-point running average.

7. Finds time of the first trough or first peak in the T3/T7 interval to the nearest sample interval (33.0 in fig. 2).

8. Determines the cross-over time at the intersection of the smoothed curve in the T3/T7 interval and the mean value of the pre-first break noise (-5.1) to the nearest 0.1 msec (36.1 msec in fig. 2).

Figure 3 is a copy of the screen display showing an example of data entered (enclosed in boxes) and results produced by the automatic first break picking program operating on the demonstration record shown on figure 1. The tabulated results list for each trace the sample standard deviation (SSD) and mean value of the pre-first break noise, the arrival time of the first trough or peak (TorP) to the nearest sample-interval time and the cross-over time beyond the first T or P to the nearest 0.1 msec for 1-to-3 thresholds.

```
AUTOMATIC FIRST BREAK PICKING OF 12-TRACE RECORD
INSERT OBSERVED-DATA MDT IN THE 4924
          CODE NAME OF MDT = AR2UT
          RECORD NO ON MDT = 4
             RECORD HEADER = 13540024
           Sample interval = 0.5
                Delay time = 0
         Pre-fb start time = 15
Est. min.fb, trace 1       = 26
Est. min fb, trace 12      = 41
 Window length (msec)      = 30
First threshold mult.      = 2
 Number of thresholds      = 3
                          -COLUMN 1-        -COLUMN 2-        -COLUMN 3-
TRACE  NOISE  NOISE       2.0 X SSD         3.0 x SSD         4.0 x SSD
 NO.    SSD   MEAN       TorP CROSS        TorP CROSS        TorP CROSS
  1     6.5   -5.1       33.0  36.0        33.0  36.0        33.0  36.0
  2     7.1   -4.9       34.0  37.3        34.0  37.3        34.0  37.3
  3     7.3   -2.0       30.0  33.1        36.5  39.8        36.5  39.8
  4     5.0   -1.3       32.0  34.4        38.5  41.5        38.5  41.5
  5     5.9  -10.5       33.0  35.9        39.0  42.4        39.0  42.4
  6     3.2    9.6       40.0  43.1        40.0  43.1        40.0  43.1
  7     2.7    5.3       41.5  44.5        41.5  44.5        41.5  44.5
  8     8.3    0.7       42.5  46.3        42.5  46.3        42.5  46.3
  9     7.8    0.9       44.0  47.8        44.0  47.8        44.0  47.8
 10     7.0   -0.9       45.5  48.7        45.5  48.7        45.5  48.7
 11     2.1    6.1       45.5  49.3        45.5  49.3        45.5  49.3
 12     2.5    8.3       47.0  51.0        47.0  51.0        47.0  51.0

DO YOU WANT TO ACCEPT ANY SET OF THE ABOVE? (Y OR N) Y
COLUMN NUMBER OF ACCEPTED DATA SET = 3
DO YOU WANT TO STORE FIRST ARRIVALS ON TAPE? (Y OR N) Y
INSERT FIRST BREAK MDT IN 4051    FILE NUMBER = 20
DATA STORED AND RETRIEVABLE
PROGRAM COMPLETED
```

Figure 3. Copy of the screen display showing an example of data entered and results produced by the automatic first break picking program. Sketched arrows (done by the author, not the program) point to arrival times that from inspection are not acceptable.

Let us trace through the entries and results displayed on the example shown on figure 3. After you enter the code name of the MDT (AR2UT) and the record number on this MDT (record number 4), the program find this record and reads and prints its header (13540024), sample interval (0.5 msec), and delay time (0.0 msec). Next you enter minimum first break times for traces 1 and 12 (26 and 41 msec respectively), the window length beyond the minimum first break time (30 msec), the first threshold multiplier (2), and the number of threshold multipliers (a maximum of 3). Of course, the amount of time required to make the computations depends of the number of thresholds requested. If, for example, your experience indicates that a single threshold multiplier is sufficient, say a value of 5, then you would enter a 5 for the first threshold multiplier and a 1 for the number of thresholds. Once these entries are made, automatic computation of first arrivals begins.

After computation and tabulation of results, you are asked if you want to accept any set of them. If you choose not to accept any of the results tabulated, then the program returns to the data-entry segment beginning with entry of the estimated minimum first-break time for trace 1. If you choose to accept the results presented, then you are prompted to enter the column number of the selected pair of arrival times--in this example, column 3 for which a threshold value of four times the sample standard deviation of the pre-first arrival noise was used. Note in this example that the first trough and cross-over times listed in column 1 contain values (indicated by sketched arrows) whose validities are difficult to accept after inspection of the original data displayed on figure 1.

Next you are asked if you want to store the selected results on a first break data tape. After entering the number of the file on which you want these data saved, the data are stored, and a read-after-write operation is performed in order to insure that the data have not only been stored but also are retrievable.

EVENT ALIGNMENT PROGRAMS

Two programs are used for event alignment: one in which arrival times are entered from the keyboard, and one in which these times are read from a first break data tape as generated in the automatic first break picking program.

Because both programs require data within a specified window for all 12 traces be stored in memory, considerable memory space must be allocated. The programs check to see if sufficient memory is available before computations proceed. If not enough memory exists, a warning is printed on the screen, and you are sent back to that part of the program at which window parameters are entered.

Insufficient memory also prohibits the output of final first break times to a data tape. Thus, the output of these programs is a hard copy of the screen displays. If your computer has more than a 32K-byte memory, I suggest that you combine the first and third programs of this report, and that you provide for storage of results.

Figure 4 shows an example of the information entered and the first screen display produced using the keyboard-entry program. After the code name of the MDT and the number of the record on that MDT are entered, the computer finds and reads the file containing this record's



Figure 4. Copy of screen display showing information entered and display produced by the event-alignment program using arrival times entered from the keyboard.

header and then prints it (13540024) and the delay time--the time at which recording begins--in this case, 0 msec. Next you are asked if you want to use all 12 traces. If an N is entered, you are asked for the beginning and ending trace numbers. If a Y is entered, you are prompted to enter the sample interval of the observed data (0.5 msec), the sample interval within the window to be used in the display (1 msec), the length of the window (50 msec), and the timing line interval (5 msec). Finally, you enter the arrival time and display amplitude multiplier for each trace. In this example, the times entered were for the first breaks as hand picked on the expanded segment of the test record; the amplitude multiplier for the initial display was entered as a 1 for each trace.

Upon entering the amplitude multiplier for the last trace (trace 12 in the example shown on fig. 4), the display shown on the bottom half of figure 4 is displayed on the screen. Note that the entered arrival times are listed in the column on the left, the entered amplitude multipliers are listed on the right, and the arrivals are shifted so that the entered times are aligned at zero time on the display. For example, the arrivals for trace 1 are shifted 25 msec to the left. After the shifted traces are displayed, you are asked--in the question printed across the bottom of the display--if you want to accept the arrival times as displayed. If you answer with a Y, the program ends at this point; if you answer with an N, the screen display remains and the flashing cursor is moved to the first row of column two--under the DEV label.

Now the eye-straining work begins. You must look at the screen and decide how much deviation to enter. The program is written such that a deviation to the right of the zero line is considered as a positive deviation, and that to the left, a negative deviation. After the deviation is entered, the cursor moves to the far right-hand column. Here you are to enter a multiplier to alter the displayed amplitude. If, for example, the initial multilier is 3 and you enter a multiplier of 2, then on subsequent displays the amplitude multiplier would be 6.

After the deviation and amplitude multiplier for the last trace are entered, you are asked (lower left-hand side of display) if you want to display the shifted traces, upper display of figure 5. If you answer with an N, then the program ends and the arrival times as listed on the left side of the display do NOT incorporate the deviations entered. If you answer with a Y, then the new plot (lower display of fig. 5) shows the adjusted first arrival times and the traces are shifted and plotted with the new set of multipliers as listed on the left side of the far right column. Again you are asked if you want to accept these arrival times. In this example, we replied with an N, and then we entered the deviations as shown in the deviations column and amplitude multipliers shown in the far right column.

Figure 6 shows the final display for the sample problem in which initial arrival times were entered from the keyboard. Here we have finally decided to accept the arrival times as listed in the left-hand column. In this example we have attempted to align first arrivals on the onset of first downward movement of the trace by using the first-crossing times (about 10 msec beyond the onset times) as a guide in determining the deviations entered in the event-alignment program.



Figure 5. Copy of screen displays showing examples of subsequent entries and results produced when using the event-alignment program in which initial arrival times are entered from the keyboard.



Figure 6. Copy of screen display showing final display of example to illustrate use of event-alignment program using arrivals times entered from the keyboard.

Let us now look at an example of results produced by the event-alignment program that uses first-arrival data stored on a tape. Figure 7 shows typical information entered in this program.

```
ALIGN ARRIVALS ON SCREEN USING INTIAL VALUES FROM MDT
INSERT FIRST-ARRIVAL MDT IN 4051    FILE NUMBER = [62]
DO YOU WANT TO USE FIRST-EXTREMUM TIMES? (Y OR N) [N]
INSERT MDT: AR2UT IN 4924
ARE YOU READY TO PROCEED? (Y OR N) [Y]
         HEADER = 13548824
     DELAY TIME = 0

     WINDOW S.I. = [0.5]
LENGTH OF WINDOW = [20]
  TIME-LINE INT. = [2]
```

Figure 7. Copy of screen display showing information entered into the event-alignment program that uses arrival-time data from a MDT.

The first-arrival data tape contains, in addition to the first-arrival times, the code number of the MDT and the number of the record; thus, these data need not be entered as they were with the previous event-alignment program. Of course, you must tell the computer in which file these data have been stored (in this example, file number 62).

Next, you are asked if you want to use first-extremum times. If you answer with a Y, then the first peaks or troughs are aligned at zero time on the display; if you answer with an N, then the cross-over times following the first extremum time are shifted so as to align at zero time on the display. Following the supplying of this answer, the computer reads data from the first-arrival data tape inserted in the 4051. It then instructs you to insert the MDT containing record data in the 4924. Upon receiving a Y response to the ready-to-proceed question, the header and delay time of the record are printed.

Finally, as in the previous event alignment program, you are asked to enter the window sample interval (0.5 msec), the length of the window (20 msec), and the timing-line interval (2 msec). Upon entry of this information, the first display (fig. 8) is produced.



Figure 8. Copy of screen display showing first display produced by the event alignment program that uses first arrivals from a data tape.

Subsequent processing with this program follows the same procedure as that of the first event-alignment program previously discussed. In this example, we chose (as indicated by the Y answer to the question at the bottom of the plot) to accept the first-arrival times as determined by the automatic first-break picking program.

ADDITIONAL COMMENTS ON THE PROGRAMS

In order to put the programs of this report to work, you must know how to perform the following operations:

1. transcribe the program into the computer,

2. store the program on magnetic tape,
3. retrieve the program from magnetic tape,
4. enter prompted information from the keyboard, and
5. copy the screen display.

These tasks are well documented in the computer's operator's manuals.

Four control characters (ones requiring the holding down of the control key as the letter is entered) are used in the programs: G (ring bell), K (move cursor up one line), L (erase screen and move cursor to the HOME position), and the RUB OUT (move cursor to the left margin and down one line). Because the printer used to list the programs cannot print an underscore, in the printed listing these control characters are shown as G_, K_, L_, and __.

To achieve maximum data packing on master data tapes, all record data are stored as three-character hexadecimal values. Special ROMs are used after reading data from a MDT to convert them from hexadecimal to digital values and later to convert digital values to hexadecimal prior to storing them on a master data tape. The occurrences of these ROMs can be recognized by statements of the form: CALL "HEXDEC",B$,V,LEN (B$),3 (where B$ is the string variable containing the data in hexadecimal and V is the array containing the data in decimal), and "DECHEX",B$,V,1001,3 (where 1001 is the size of the V array).

REFERENCES

Anstey, N. A., 1977, Seismic interpretation: the physical aspects: International Human Resources Development Corporation, Boston, Mass., 637 p.

Ricker, Norman, 1953, The form and laws of propagation of seismic wavelets: Geophysics, vol. 18, no. 1, p. 10-40.

AUTOMATIC FIRST ARRIVAL PICKING PROGRAM

```
100 PRINT "L_AUTOMATIC FIRST BREAK PICKING OF 12-TRACE RECORD"
110 INIT
120 REM ** ENTER CODE NAME AND RECORD NUMBER
130 GOSUB 520
140 REM ** FIND, RETRIEVE, AND DECODE HEADER FILE
150 GOSUB 680
160 REM ** ENTER INPUT PARAMETERS
170 GOSUB 910
180 REM ** MAKE REQUIRED CALCULATIONS FOR EACH TRACE
190 M=R1-1
200 FOR N=1 TO 12
210 M=M+1
220 GOSUB 250
230 NEXT N
240 GO TO 460
250 REM ** SUB: COMMON PROCEDURE FOR ALL TRACES
260 REM ** FIND, RETRIEVE, CONVERT, AND SCALE DATA
270 GOSUB 1180
280 REM ** DETERMINE PRE-FB NOISE LEVEL
290 GOSUB 1310
300 L2=L3-1
310 FOR Q=1 TO Q3
320 L2=L2+1
330 X4=L2*X3
340 REM ** LOCATE TIME (T3) AT WHICH ABS AMPL > L2*SSD
350 GOSUB 1520
360 REM ** LOCATE SECOND CROSS-OVER TIME (T7) BEYOND T3
370 GOSUB 1600
380 REM ** SMOOTH BETWEEN T3 & T7 USING 7-PT RUNNING AVERAGE
390 GOSUB 1740
400 REM ** LOCATE TIMES (T4&T6) OF EXTREMA BETWEEN T3 AND T7
410 GOSUB 1890
420 REM ** LOCATE CROSS-OVER TIME (T5) BETWEEN T4 AND T6
430 GOSUB 1980
440 NEXT Q
450 RETURN
460 REM ** TABULATE FIRST ARRIVALS
470 GOSUB 2100
480 REM ** STORE FIRST ARRIVALS
490 GOSUB 2530
500 PRINT "G_G_G_PROGRAM COMPLETED"
510 END
520 REM ** SUB: ENTER CODE NAME AND RECORD NUMBER
530 DIM B$(3005),C$(18),D$(18),G$(1),H$(11),I$(1),L$(2),M$(40),N$(5)
540 DIM J0(12),J9(12),S2(12),T0(12),T4(12,3),T5(12,3),X5(12)
550 M$="FIRST ARRIVALS USING AUTOMATIC PICKING"
560 S2=0
570 T4=0
580 T5=0
590 X5=0
600 PRINT "G_G_G_INSERT OBSERVED-DATA MDT IN THE 4924"
610 PRINT "     CODE NAME OF MDT = ";
620 INPUT N$
630 PRINT "     RECORD NO ON MDT = ";
640 INPUT R
650 R1=R*12-10
660 R2=R*12+1
670 RETURN
680 REM ** SUB: FIND, RETRIEVE, AND DECODE HEADER FILE
690 FIND @2:R1
700 READ @2:H$
710 PRINT "        RECORD HEADER = ";H$
720 I$=SEG(H$,4,1)
730 L$=SEG(H$,5,2)
740 S1=VAL(I$)
```

```
750 L=VAL(L$)*10
760 GO TO S1 OF 770,790,810,830,850,870
770 S1=0.05
780 GO TO 880
790 S1=0.1
800 GO TO 880
810 S1=0.2
820 GO TO 880
830 S1=0.5
840 GO TO 880
850 S1=1
860 GO TO 880
870 S1=2
880 PRINT "      Sample interval = ";S1
890 PRINT "           Delay time = ";L
900 RETURN
910 REM ** SUB: ENTER INPUT PARAMETERS
920 PRINT "    Pre-fb start time = ";
930 INPUT T9
940 J1=1+T9/S1
950 PRINT "Est. min.fb, trace 1  = ";
960 INPUT T0(1)
970 J0(1)=(T0(1)-L)/S1+1
980 PRINT "Est. min fb, trace 12 = ";
990 INPUT T0(12)
1000 J0(12)=(T0(12)-L)/S1+1
1010 K0=(J0(12)-J0(1))/11
1020 FOR J=2 TO 11
1030 J0(J)=J0(J-1)+K0
1040 NEXT J
1050 J0=J0+0.5
1060 J0=INT(J0)
1070 T0=J0-1
1080 T0=T0*S1
1090 PRINT " Window length (msec) = ";
1100 INPUT L8
1110 J8=L8/S1
1120 J9=J0+J8
1130 PRINT "First threshold mult. = ";
1140 INPUT L3
1150 PRINT " Number of thresholds = ";
1160 INPUT Q3
1170 RETURN
1180 REM ** SUB: FIND, RETRIEVE, CONVERT, AND SCALE DATA
1190 DELETE V
1200 DIM V(J9(N))
1210 FIND @2:M
1220 IF M=R1 THEN 1250
1230 READ @2:B$
1240 GO TO 1260
1250 READ @2:H$,B$
1260 B$=SEG(B$,1,3*J9(N))
1270 V=0
1280 CALL "HEXDEC",B$,V,LEN(B$),3
1290 V=V-511
1300 RETURN
1310 REM ** SUB: DETERMINE PRE-FB NOISE LEVEL
1320 DELETE V1,X2
1330 C=J0(N)-J1+1
1340 DIM V1(C),X2(C)
1350 K=0
1360 FOR J=J1 TO J0(N)
1370 K=K+1
1380 V1(K)=V(J)
1390 NEXT J
```

```
1400 X1=SUM(V1)
1410 X1=X1/C
1420 X5(N)=X1
1430 X2=V1-X1
1440 X2=X2↑2
1450 X3=SUM(X2)
1460 X3=SQR(X3/C)
1470 S2(N)=X3
1480 FOR J=J0(N) TO J9(N)
1490 V(J)=V(J)-X1
1500 NEXT J
1510 RETURN
1520 REM ** SUB: LOCATE TIME (T3) AT WHICH ABS AMPL > L2*SSD
1530 T3=T0(N)-S1
1540 FOR J=J0(N) TO J9(N)
1550 T3=T3+S1
1560 IF ABS(V(J))>X4 THEN 1580
1570 NEXT J
1580 J3=(T3-L)/S1+1
1590 RETURN
1600 REM ** SUB: LOCATE SECOND CROSS-OVER TIME (T7) BEYOND T3
1610 DELETE V2
1620 DIM V2(J9(N)-J3+1)
1630 K=0
1640 FOR J=J3 TO J9(N)
1650 K=K+1
1660 V2(K)=V(J)
1670 NEXT J
1680 CALL "CROSS",V2,0,J7,2
1690 DELETE V2
1700 J7=INT(J7+0.5)
1710 T7=T3+(J7-1)*S1
1720 J7=(T7-L)/S1+1
1730 RETURN
1740 REM ** SUB: SMOOTH BETWEEN T3 & T7 USING 7-PT RUNNING AVERAGE
1750 N3=J7-J3
1760 N7=N3+1
1770 DELETE W1
1780 DIM W1(N7)
1790 P=0
1800 FOR J=J3 TO J7
1810 S4=0
1820 FOR K=1 TO 7
1830 S4=S4+V(J+K-4)
1840 NEXT K
1850 P=P+1
1860 W1(P)=S4/7
1870 NEXT J
1880 RETURN
1890 REM ** SUB: LOCATE TIMES (T4&T6) OF EXTREMA BETWEEN T3 AND T7
1900 CALL "MIN",W1,M1,I1
1910 CALL "MAX",W1,M2,I2
1920 I3=I1 MIN I2
1930 I4=I1 MAX I2
1940 T4(N,Q)=T3+(I3-1)*S1
1950 T6=T3+(I4-1)*S1
1960 J4=(T4(N,Q)-L)/S1+1
1970 RETURN
1980 REM ** REM: LOCATE CROSS-OVER TIME (T5) BETWEEN T4 AND T6
1990 I5=I4-I3+1
2000 DELETE W2
2010 DIM W2(I5)
2020 K=0
2030 FOR J=I3 TO I4
2040 K=K+1
```

```
2050 W2(K)=W1(J)
2060 NEXT J
2070 CALL "CROSS",W2,0,J5
2080 T5(N,Q)=T4(N,Q)+J5*S1
2090 RETURN
2100 REM ** SUB: TABULATE VALUES
2110 IMAGE 16X,3(7X,"-COLUMN ",D,"-")
2120 L4=1
2130 PRINT USING 2110:L4,L4+1,L4+2
2140 IMAGE "TRACE  NOISE  NOISE",5X,D.D," X SSD",2(8X,D.D," x SSD")
2150 PRINT USING 2140:L3,L3+1,L3+2
2160 IMAGE" NO.     SSD   MEAN",5X,"TorP CROSS",2(7X,"TorP CROSS")
2170 PRINT USING 2160:
2180 IMAGE X,2D,3X,3D.D,2X,3D.D
2190 IMAGE "K_",22X,3D.D,X,3D.D
2200 IMAGE "K_",39X,3D.D,X,3D.D
2210 IMAGE "K_",56X,3D.D,X,3D.D
2220 FOR N=1 TO 12
2230 PRINT USING 2180:N,S2(N),X5(N)
2240 PRINT USING 2190:T4(N,1),T5(N,1)
2250 PRINT USING 2200:T4(N,2),T5(N,2)
2260 PRINT USING 2210:T4(N,3),T5(N,3)
2270 NEXT N
2280 PRINT "__DO YOU WANT TO ACCEPT ANY SET OF THE ABOVE? (Y OR N) ";
2290 INPUT G$
2300 IF G$="Y" THEN 2330
2310 PAGE
2320 GO TO 160
2330 PRINT "COLUMN NUMBER OF ACCEPTED DATA SET = ";
2340 INPUT C0
2350 DELETE T9
2360 DIM T8(12),T9(12)
2370 GO TO C0 OF 2380,2430,2480
2380 FOR K=1 TO 12
2390 T8(K)=T4(K,1)
2400 T9(K)=T5(K,1)
2410 NEXT K
2420 RETURN
2430 FOR K=1 TO 12
2440 T8(K)=T4(K,2)
2450 T9(K)=T5(K,3)
2460 NEXT K
2470 RETURN
2480 FOR K=1 TO 12
2490 T8(K)=T4(K,3)
2500 T9(K)=T5(K,3)
2510 NEXT K
2520 RETURN
2530 REM ** SUB: STORE FIRST ARRIVALS
2540 PRINT "DO YOU WANT TO STORE FIRST ARRIVALS ON TAPE? (Y OR N) ";
2550 INPUT G$
2560 IF G$="N" THEN 2630
2570 PRINT "G_G_G_INSERT FIRST BREAK MDT IN 4051   FILE NUMBER = ";
2580 INPUT F9
2590 N0=12
2600 FIND F9
2610 WRITE N0,M$,N$,R,S1,L,T8,T9
2620 CLOSE
2630 DELETE N0,M$,N$,R,S1,L,T8,T9
2640 FIND F9
2650 READ @33:N0,M$,N$,R,S1,L
2660 DIM T8(N0),T9(N0)
2670 READ @33:T8,T9
2680 PRINT "DATA STORED AND RETRIEVABLE"
2690 RETURN
```

EVENT ALIGNMENT PROGRAM USING DATA FROM KEYBOARD

```
100 PRINT "L_ALIGN ARRIVALS ON SCREEN USING INTIAL VALUES FROM KEYBOARD"
110 INIT
120 DATA 6,96,90,1,50,91,1001
130 READ B1,B2,B3,W3,W9,W0,N1
140 DIM D$(3005),G$(1),H$(11),M$(5),L$(7),O(N1)
150 PRINT "G_G_G___INSERT MDT IN 4924"
160 PRINT "__CODE NAME OF MDT = ";
170 INPUT M$
180 PRINT "RECORD NO ON MDT = ";
190 INPUT R1
200 F9=(R1-1)*12+2
210 FIND @2:F9
220 READ @2:H$
230 L$=SEG(H$,5,2)
240 T0=VAL(L$)
250 REM ** ESTABLISH INITIAL VALUES
260 GOSUB 480
270 REM ** FILL WINDOW ARRAY
280 GOSUB 970
290 REM ** CREATE SCREEN DISPLAY
300 GOSUB 1180
310 REM ** PLOT SHIFTED TRACE SEGMENTS
320 GOSUB 1610
330 WINDOW 0,130,0,100
340 VIEWPORT 0,130,0,100
350 MOVE 0,3
360 PRINT "DO YOU WANT TO ACCEPT THESE ARRIVAL TIMES? (Y OR N) ";
370 INPUT G$
380 IF G$="N" THEN 400
390 GO TO 460
400 REM ** ENTER NEW TIME SHIFTS AND AMPLITUDES
410 GOSUB 1760
420 PRINT "__DISPLAY SHIFTED TRACES? ";
430 INPUT G$
440 IF G$="N" THEN 460
450 GO TO 290
460 PRINT "G_G_G_PROGRAM COMPLETED"
470 END
480 REM ** SUB: ESTABLISH INITIAL FB WINDOW
490 PRINT "          HEADER = ";H$
500 PRINT "       DELAY TIME = ";T0
510 PRINT "__DO YOU WANT ALL 12 TRACES? (Y OR N) ";
520 INPUT G$
530 IF G$="Y" THEN 590
540 PRINT "First trace number = ";
550 INPUT R2
560 PRINT " Last trace number = ";
570 INPUT R3
580 GO TO 610
590 R2=1
600 R3=12
610 N3=R3-R2+1
620 PRINT "__  OBS. DATA S.I. = ";
630 INPUT S1
640 PRINT "     WINDOW S.I. = ";
650 INPUT S2
660 S3=S2/S1
670 PRINT "LENGTH OF WINDOW = ";
680 INPUT W1
```

```
690 N2=W1/S2+1
700 N4=N2+10/S2
710 PRINT "  TIME-LINE INT. = ";
720 INPUT W2
730 W3=INT((W1+10)/W2)
740 DELETE J1,K1,S4,T1,T2
750 DIM J1(N3),K1(N3),S4(N3),T1(N3),T2(N3)
760 J1=0
770 K1=0
780 S4=0
790 T1=0
800 T2=0
810 M0=120*N4
820 IF M0<MEMORY THEN 850
830 PRINT "G_G_G___INSUFFICIENT MEMORY"
840 GO TO 510
850 PRINT "G_G_G___NOTE: ARRIVAL TIME MUST BE =>10"
860 PRINT "__TRACE  ARRIVAL TIME   AMPL. MULT."
870 R4=R2-1
880 FOR J=1 TO N3
890 R4=R4+1
900 PRINT "  ";R4;"
910 INPUT T1(J)
920 J1(J)=(T1(J)-T0)/S1+1
930 PRINT "K_
940 INPUT K1(J)
950 NEXT J
960 RETURN
970 REM ** SUB: FILL WINDOW ARRAY
980 DIM V(N3,N4)
990 V=0
1000 O=511
1010 J2=W1/S1
1020 FOR J=1 TO N3
1030 M=J+R2-1
1040 FIND @2:F9+M-1
1050 IF M>1 THEN 1080
1060 READ @2:H$,D$
1070 GO TO 1090
1080 READ @2:D$
1090 CALL "HEXDEC",D$,O,3003,3
1100 O=O-511
1110 K=0
1120 FOR I=J1(J)-10/S1 TO J1(J)+W1/S1 STEP S3
1130 K=K+1
1140 V(J,K)=O(I)
1150 NEXT I
1160 NEXT J
1170 RETURN
1180 REM ** SUB: CREATE SCREEN DISPLAY
1190 PAGE
1200 MOVE 0,100
1210 PRINT "DATA FROM MDT:";M$;",RECORD ";R1;",TRACES ";R2;"-";R3;",";
1220 PRINT "HEADER ";H$;", WINDOW SI=";S2
1230 D1=B3/(R3-R2+4)
1240 K2=88/W3
1250 K4=88/(W1+10)
1260 MOVE 0,B2
1270 PRINT "ARR TIME  DEV
1280 PRINT "             AMPL"
1290 IMAGE  3D.D,59X,2D.D
```

```
1300 MOVE 0,B2
1310 RDRAW 130,0
1320 RDRAW 0,-B3
1330 RDRAW -130,0
1340 RDRAW 0,B3
1350 RMOVE 14,0
1360 RDRAW 0,-B3
1370 RMOVE 14,0
1380 RDRAW 0,B3
1390 RMOVE -28,-2*D1
1400 FOR J=1 TO N3
1410 RDRAW 130,0
1420 RMOVE -130,0
1430 PRINT USING 1290:T1(J),K1(J)
1440 RMOVE 0,-D1
1450 NEXT J
1460 MOVE 25.5,B2
1470 W4=2*W2
1480 T5=-10-W4
1490 FOR J=1 TO W3+1 STEP 2
1500 T5=T5+W4
1510 PRINT T5
1520 RMOVE 2*K2,0
1530 NEXT J
1540 MOVE 28,B2
1550 FOR J=1 TO W3
1560 RMOVE K2,0
1570 RDRAW 0,-B3
1580 RMOVE 0,B3
1590 NEXT J
1600 RETURN
1610 REM ** SUB: PLOT SHIFTED TRACE SEGMENTS
1620 DIM A(N4)
1630 A=0
1640 WINDOW 1,N4,-2047,2048
1650 D6=B2-4*D1
1660 FOR J=1 TO N3
1670 D7=D6+4*D1
1680 VIEWPORT 28-S4(J),116-S4(J),D6,D7
1690 FOR K=1 TO N4
1700 A(K)=K1(J)*V(J,K)
1710 NEXT K
1720 CALL "DISP",A
1730 D6=D6-D1
1740 NEXT J
1750 RETURN
1760 REM ** SUB: ENTER NEW TIME SHIFTS AND AMPLITUDES
1770 MOVE 16,B2-D1
1780 FOR J=1 TO N3
1790 RMOVE 0,-D1
1800 PRINT " ";
1810 INPUT T3
1820 T2(J)=T2(J)+T3
1830 S4(J)=T2(J)*K4
1840 T1(J)=T1(J)+T3
1850 PRINT "K_
1860 PRINT "
1870 INPUT K3
1880 K1(J)=K1(J)*K3
1890 NEXT J
1900 RETURN
```

EVENT ALIGNMENT PROGRAM USING DATA FROM TAPE

```
100 PRINT "L_ALIGN ARRIVALS ON SCREEN USING INTIAL VALUES FROM MDT"
110 INIT
120 DATA 6,96,90,1,50,91,1001
130 READ B1,B2,B3,W3,W9,W0,N1
140 DIM D$(3005),G$(1),H$(11),M$(5),N$(50),O(N1)
150 PRINT "G_G_G___INSERT FIRST-ARRIVAL MDT IN 4051   FILE NUMBER = ";
160 INPUT F0
170 FIND F0
180 READ @33:N0,N$,M$,R1,S1,T0
190 DIM J1(12),K1(12),S4(12),T1(12),T2(12),T3(12),T4(12)
200 READ @33:T3,T4
210 PRINT "__DO YOU WANT TO USE FIRST-EXTREMUM TIMES? (Y OR N) ";
220 INPUT G$
230 IF G$="N" THEN 260
240 T1=T3
250 GO TO 270
260 T1=T4
270 DELETE T3,T4
280 J1=T1-T0
290 J1=J1/S1
300 J1=J1+1
310 K1=4
320 S4=0
330 T2=0
340 PRINT "G_G_G___INSERT MDT: ";M$;" IN 4924"
350 GOSUB 1730
360 F9=(R1-1)*12+2
370 FIND @2:F9
380 READ @2:H$
390 REM ** ESTABLISH INITIAL VALUES
400 GOSUB 620
410 REM ** FILL WINDOW ARRAY
420 GOSUB 800
430 REM ** CREATE SCREEN DISPLAY
440 GOSUB 1000
450 REM ** PLOT SHIFTED TRACE SEGMENTS
460 GOSUB 1430
470 WINDOW 0,130,0,100
480 VIEWPORT 0,130,0,100
490 MOVE 0,3
500 PRINT "DO YOU WANT TO ACCEPT THESE ARRIVAL TIMES? (Y OR N) ";
510 INPUT G$
520 IF G$="N" THEN 540
530 GO TO 600
540 REM ** ENTER NEW TIME SHIFTS AND AMPLITUDES
550 GOSUB 1580
560 PRINT "__DISPLAY SHIFTED TRACES? ";
570 INPUT G$
580 IF G$="N" THEN 600
590 GO TO 430
600 PRINT "G_G_G_PROGRAM COMPLETED"
610 END
620 REM ** SUB: ESTABLISH INITIAL FB WINDOW
630 PRINT "__            HEADER = ";H$
640 PRINT "         DELAY TIME = ";T0
650 PRINT "__        WINDOW S.I. = ";
```

```
660 INPUT S2
670 S3=S2/S1
680 PRINT "LENGTH OF WINDOW = ";
690 INPUT W1
700 N2=W1/S2+1
710 N4=N2+10/S2
720 PRINT "  TIME-LINE INT.  = ";
730 INPUT W2
740 W3=INT((W1+10)/W2)
750 M0=120*N4
760 IF M0<MEMORY THEN 790
770 PRINT "G_G_G___INSUFFICIENT MEMORY"
780 GO TO 650
790 RETURN
800 REM ** SUB: FILL WINDOW ARRAY
810 DIM V(12,N4)
820 V=0
830 O=511
840 J2=W1/S1
850 FOR J=1 TO 12
860 FIND @2:F9+J-1
870 IF J>1 THEN 900
880 READ @2:H$,D$
890 GO TO 910
900 READ @2:D$
910 CALL "HEXDEC",D$,O,3003,3
920 O=O-511
930 K=0
940 FOR I=J1(J)-10/S1 TO J1(J)+W1/S1 STEP S3
950 K=K+1
960 V(J,K)=O(I)
970 NEXT I
980 NEXT J
990 RETURN
1000 REM ** SUB: CREATE SCREEN DISPLAY
1010 PAGE
1020 MOVE 0,100
1030 PRINT "AUTO(FILE:";F0;")--DATA ON MDT:";M$;",RECORD ";R1;",";
1040 PRINT "HEADER ";H$;", WINDOW SI=";S2
1050 D1=B3/15
1060 K2=88/W3
1070 K4=88/(W1+10)
1080 MOVE 0,B2
1090 PRINT "ARR TIME  DEV
1100 PRINT "              AMPL"
1110 IMAGE  3D.D,59X,2D.D
1120 MOVE 0,B2
1130 RDRAW 130,0
1140 RDRAW 0,-B3
1150 RDRAW -130,0
1160 RDRAW 0,B3
1170 RMOVE 14,0
1180 RDRAW 0,-B3
1190 RMOVE 14,0
1200 RDRAW 0,B3
1210 RMOVE -28,-2*D1
```

```
1220 FOR J=1 TO 12
1230 RDRAW 130,0
1240 RMOVE -130,0
1250 PRINT USING 1110:T1(J),K1(J)
1260 RMOVE 0,-D1
1270 NEXT J
1280 MOVE 25.5,B2
1290 W4=2*W2
1300 T5=-10-W4
1310 FOR J=1 TO W3+1 STEP 2
1320 T5=T5+W4
1330 PRINT T5
1340 RMOVE 2*K2,0
1350 NEXT J
1360 MOVE 28,B2
1370 FOR J=1 TO W3
1380 RMOVE K2,0
1390 RDRAW 0,-B3
1400 RMOVE 0,B3
1410 NEXT J
1420 RETURN
1430 REM ** SUB: PLOT SHIFTED TRACE SEGMENTS
1440 DIM A(N4)
1450 A=0
1460 WINDOW 1,N4,-2047,2048
1470 D6=B2-4*D1
1480 FOR J=1 TO 12
1490 D7=D6+4*D1
1500 VIEWPORT 28-S4(J),116-S4(J),D6,D7
1510 FOR K=1 TO N4
1520 A(K)=K1(J)*V(J,K)
1530 NEXT K
1540 CALL "DISP",A
1550 D6=D6-D1
1560 NEXT J
1570 RETURN
1580 REM ** SUB: ENTER NEW TIME SHIFTS AND AMPLITUDES
1590 MOVE 16,B2-D1
1600 FOR J=1 TO 12
1610 RMOVE 0,-D1
1620 PRINT " ";
1630 INPUT T3
1640 T2(J)=T2(J)+T3
1650 S4(J)=T2(J)*K4
1660 T1(J)=T1(J)+T3
1670 PRINT "K_
1680 PRINT "
1690 INPUT K3
1700 K1(J)=K1(J)*K3
1710 NEXT J
1720 RETURN
1730 REM ** SUB: READY
1740 PRINT "__ARE YOU READY TO PROCEED? (Y OR N) ";
1750 INPUT G$
1760 IF G$="N" THEN 1740
1770 RETURN
```